
Aki Taanila

EXCEL RAHOITUSFUNKTIOITA

25.3.2011

SISÄLLYS

0 JOHDANTO	1
1 AIKAAN LIITTYVIÄ LASKELMIA.....	2
2 EXCELIN RAHOITUSFUNKTIOITA	3
2.1 Koronkorkolaskenta.....	3
2.2 Jaksolliset suoritukset	5
2.3 Luotot.....	8
2.4 Investointilaskelmat	11

0 JOHDANTO

Viimeisin versio tästä monisteesta ja siihen liittyvästä materiaalista löytyy osoitteesta <http://myy.haaga-helia.fi/~taaak/f>

Tämän monisteen alussa tarkastellaan aikaan liittyviä laskelmia. Tämän jälkeen käsitellään yleisimmin käytetyt rahoitusfunktiot.

VAROITUS! Excel on hyvä työväline liike-elämän laskelmien suorittamiseen. Edellytyksenä on, että laskelmien tekijä hallitsee laskentamenetelmien matemaattisen perustan. Matemaattisen perustan oppimiseen Excel ei sovellu. Tämän sivuston lukijan oletetaan hallitsevan esitettyihin menetelmiin liittyvän matemaattisen perustan (suositeltava kirja: Saaranen & Koltola & Pösö. Liike-elämän matematiikkaa. Edita).

Dokumentissa esiintyy viittauksia Excel-esimerkkeihin, joihin tutustuminen on olennainen osa asioiden oppimista. Tässä dokumentissa viitatus Excel-esimerkit löytyvät verkosta:

- <http://myy.haaga-helia.fi/~taaak/f/aika.xlsx> (aikaan liittyviä laskelmia)
- <http://myy.haaga-helia.fi/~taaak/f/rahoi.xls> (esimerkkejä rahoitusfunktioiden käytöstä).

Tutustu myös muihin oppimateriaaleihini <http://myy.haaga-helia.fi/~taaak>

1 AIKAAN LIITTYVIÄ LASKELMIA

Voit kirjoittaa Exceliin päivämäärämuotoista tietoa esimerkiksi 23.4.2008 ja Excel esittää tiedon soluun määritetyn muotoilun mukaisesti. Excel tulkitsee päivämäärät järjestysnumeroiksi siten että 1.1.1900 vastaa järjestysnumeroa 1. Tämän mukaisesti esimerkiksi päivämäärää 23.4.2008 vastaa järjestysnumero 39561. Excelin päiväysjärjestelmässä vuosi 1900 lasketaan virheellisesti karkausvuodeksi, joten 28.2.1900 järjestysnumero on 59 ja seuraavan päivän 1.3.1900 järjestysnumero on 61. Vuoden 1900 jälkeiset karkausvuodet Excel huomioi oikein.

Tärkeä huomautus: Seuraavassa esitellään joitain funktioita, joiden argumentteina voidaan käyttää viittauksia päivämääriä sisältäviin soluihin. Excel muuntaa päivämääriin osoittavat viittaukset järjestysnumeroiksi. Päivämääriä ei voi sellaisenaan kirjoittaa funktion argumentiksi, koska tällöin Excel odottaa argumenteiksi järjestysnumeroita.

Koska päivämäärät ovat itse asiassa järjestysnumeroita, niin päivämääriä voidaan vähentää toisistaan ja laskea kahden päivämäärän välinen aika päivinä.

- Kahden päivämäärän väliset todelliset päivät (englantilainen tapa) saadaan suoraan vähennyslaskuna. Jos vähennyslaskun tulos on muotoiltu päivämääräksi, niin vaihda siihen tavallinen luku-muotoilu.
- Saksalaisen tavan mukaan lasketut päivät saadaan funktiolla **=PÄIVÄT360(alkupvm;loppupvm;1)** (engl. **DAYS360**). Saksalaisessa tavassa kaikki kuukaudet lasketaan pituudeltaan 30 päivän mittaisiksi. Jos alkupvm tai loppupvm on kuukauden 31. päivä, niin funktio tulkitsee sen 30. päiväksi.
- Amerikkalaisen NASD-menetelmän mukaan lasketut päivät saadaan funktiolla **=PÄIVÄT360(alkupvm;loppupvm)**. NASD-menetelmä poikkeaa saksalaisesta, jos loppupvm on kuukauden 31. päivä (Jos loppupvm on kuukauden 31. päivä ja alkupvm on aiempi kuin kuukauden 30. päivä, loppupvm:stä tulee seuraavan kuukauden 1. päivää vastaava päivä. Muussa tapauksessa loppupvm:stä tulee sen kuukauden 30. päivää vastaava päivä.).
- Työpäivät 5-päiväisen työviikon mukaisesti saadaan funktiolla **=TYÖPÄIVÄT(alkupvm;loppupvm)** (engl. **NETWORKDAYS**).
- **=TYÖPÄIVÄ(alkupvm;päiviä)** (engl. **WORKDAY**) lisää alkupvm:ään sen verran työpäiviä (viikonloput siis hypätään yli) kuin päiviä argumentti osoittaa.

Excel osaa myös kellonajat. Kellonaika lisätään päivämäärää vastaavaan järjestyslukuun desimaaliosana. Esimerkiksi 23.4.2008 12:00 vastaa desimaalilukua 39561,5. Jos syötät kellonajan ilman päivämäärää ja vuosilukua, niin Excel olettaa että kyseessä on 0.1.1900. Laskutoimituksissa on odotettavissa ongelmia, jos joudut 1800-luvun puolelle. Ongelmat ovat seurausta siitä, että Excelin päivämääräjärjestelmä ei ulotu vuotta 1900 kauemmaksi menneisyyteen.

Nykyhetken laskemisen voit suorittaa seuraavilla tavoilla:

- Funktio **=TÄMÄ.PÄIVÄ()** (engl. **TODAY**) palauttaa kuluvan päivän päivämäärän. Taulukon uudelleenlaskennan yhteydessä funktion antama tulos päivittyy.
- Funktio **=NYT()** (engl. **NOW**) palauttaa sen hetkisen kellonajan. Taulukon uudelleenlaskennan yhteydessä funktion antama tulos päivittyy.

Excel esimerkeistä <http://myy.haaga-helia.fi/~taaak/f/aika.xlsx> opit lisää ja löydät mielenkiintoisia esimerkkejä.

2 EXCELIN RAHOITUSFUNKTIOITA

Yksi taulukkolaskennan perusideoista on lähtötietojen kirjoittaminen omiin soluihinsa. Näin saadaan laskentamalleja, joiden avulla voidaan vaivatta tarkastella tilannetta erilaisilla lähtötiedoilla. Kaikki tämän dokumentin esimerkit on laadittu laskentamallien muotoon.

Seuraavassa on esitelty yleisimmin käytetyt Excelin rahoitusfunktiot. Löydät esitetyt esimerkit laskettuina tiedostosta <http://myy.haaga-helia.fi/~taaak/f/rahoi.xls>.

2.1 Koronkorkolaskenta

Koronkorkolaskennassa tarkastellaan kestoltaan useamman korkojakson mittaisia aikavälejä. Korkojaksona on usein vuosi, mutta korkojakso voi olla muunkin pituinen (esimerkiksi puoli vuotta, neljännesvuosi tai kuukausi). Koronkorkolaskelmissa on aina käytettävä nimenomaan korkojaksoon liittyvää korkokantaa. Koronkorkolaskelmissa on neljä muuttujaa: korkokanta, korkojaksojen lukumääränä, alkuarvo ja kasvanut arvo.

Huomaa, että alkuarvon ja kasvaneen arvon täytyy funktion lähtötietoina olla erimerkkiset. Alkuarvon ja kasvaneen arvon erimerkkisyys on ymmärrettävää. Jos esimerkiksi talletan rahaa pankkiin (alkuarvo), niin raha on minulta pois (miinusta) ja nostaessani rahat korkoineen (kasvanut arvo), niin se on minulle tuloa (plussaa). Pankin kannalta tarkasteltuna talletushetkellä pankki saa talletuksen käyttöönsä (plussaa), mutta talletuksen nostohetkellä pankki menettää käytössään olleen rahamäärän (miinusta).

Alkuarvo

Alkuarvo voidaan laskea funktiolla **NA** (engl. **PV**). Kuvasta näet solussa B6 käytetyn funktion.

	A	B	C
1	Koronkorkolaskenta - alkuarvo		
2			
3	Jakson korkokanta	4,00 %	
4	Korkojaksoja	5	
5	Kasvanut arvo	15 000,00	
6	Alkuarvo	-12 328,91	=NA(B3;B4;;B5)
7			(engl. PV)

Huomaa, että funktion kolmatta argumenttia käytetään jaksollisten suoritusten yhteydessä ja tässä se on jätetty tyhjäksi (kaksi peräkkäistä puolipistettä). Laskelman mukaan tarvitaan 12328,91 euron talletus, jotta viiden vuoden kuluttua olisi korkoineen 15000 euroa.

Kasvanut arvo

Kasvanut arvo voidaan laskea funktiolla **TULEVA.ARVO** (engl. **FV**). Kuvasta näet solussa B6 käytetyn funktion.

	A	B	C
1	Koronkorkolaskenta - kasvanut arvo		
2			
3	Jakson korkokanta	7,00 %	
4	Korkojaksoja	3	
5	Alkuarvo	-10 000,00	
6	Kasvanut arvo	12 250,43	=TULEVA.ARVO(B3;B4;;B5)
7			<i>(engl. FV)</i>

Huomaa, että funktion kolmatta argumenttia käytetään jaksollisten suoritusten yhteydessä ja tässä se on jätetty tyhjäksi (kaksi peräkkäistä puolipistettä). Laskelman mukaan 10000 euron talletus kasvaa 3 vuodessa 12250,43 euroksi.

Korkokanta

Korkokanta voidaan laskea funktiolla **KORKO** (engl. **RATE**). Kuvasta näet solussa B6 käytetyn funktion.

	A	B	C
1	Koronkorkolaskenta - korkokanta		
2			
3	Korkojaksoja	5	
4	Alkuarvo	-12 000,00	
5	Kasvanut arvo	17 000,00	
6	Jakson korkokanta	7,21 %	=KORKO(B3;;B4;B5)
7			<i>(engl. RATE)</i>

Huomaa, että funktion toista argumenttia käytetään jaksollisten suoritusten yhteydessä ja tässä se on jätetty tyhjäksi (kaksi peräkkäistä puolipistettä). Laskelman mukaan tarvitaan 7,21% korkokanta, jotta 12000 euroa kasvaa 5 vuodessa 17000 euroksi.

Korkojaksojen lukumäärä

Korkojaksojen lukumäärä voidaan laskea funktiolla **NJAKSO** (engl. **NPER**). Kuvassa näet solussa B6 käytetyn funktion.

	A	B	C
1	Koronkorkolaskenta - korkojaksojen lukumäärä		
2			
3	Jakson korkokanta	8,00 %	
4	Alkuarvo	-17 000,00	
5	Kasvanut arvo	25 000,00	
6	Korkojaksoja	5,01	=NJAKSO(B3;;B4;B5)
7			<i>(engl. NPER)</i>

Huomaa, että funktion toista argumenttia käytetään jaksollisten suoritusten yhteydessä ja tässä se on jätetty tyhjäksi (kaksi peräkkäistä puolipistettä). Laskelman mukaan tarvitaan yli 5 vuotta, jotta 17000 euroa kasvaa 25000 euroksi.

2.2 Jaksolliset suoritukset

Huomaa, että funktioita käytettäessä suoritusten alkuarvon/loppuarvon täytyy olla erimerkkinen suoritusten kanssa. Jos sekä alkuarvo että loppuarvo asetetaan nolosta poikkeavaksi, niin merkkien kanssa täytyy olla erityisen tarkkana ja asettaa ne tilanteeseen sopivasti. Jos esimerkiksi alkuarvona on lainan käteisarvo ja loppuarvona lainasta maksettava jäänyt osuus, niin alkuarvo ja loppuarvo ovat keskenään erimerkkiset, mutta suoritukset ovat loppuarvon kanssa samanmerkkisiä.

Ole huolellinen suoritusten ajankohdan kanssa. Suoritukset voivat tapahtua jakson lopussa (funktioiden oletusarvo) tai jakson alussa. Eron selventämiseksi tarkastellaan yksinkertaista esimerkkiä. Jos korkokanta on 5% ja talletetaan 100 euroa kolme kertaa aina vuoden lopussa, niin viimeisen talletuksen tapahduttua loppuarvo on $1,05^2 * 100 + 1,05 * 100 + 100 = 315,25$ euroa. Jos talletukset tapahtuvat jakson alussa, niin viimeisen talletuksen tapahduttua loppuarvo on $1,05^3 * 100 + 1,05^2 * 100 + 1,05 * 100 = 331,01$ euroa.

Molemmissa tapauksissa korkokanta, jaksojen lukumäärä ja jaksollisen suorituksen suuruus ovat yhtä suuria. Ero syntyy suoritusten ajankohdasta. Excelin funktioissa viimeinen argumentti osoittaa suoritusten ajankohdan (0=jakson lopussa, 1=jakson alussa).

Jaksollisten suoritusten alkuarvo

Jaksollisten suoritusten alkuarvo voidaan laskea funktiolla **NA** (engl. **PV**). Kuvassa näet solussa B7 käytetyn funktion.

	A	B	C
1	Jaksollisten suoritusten alkuarvo		
2			
3	Jakson korkokanta	1,00 %	
4	Jaksoja	24	
5	Suoritus	500,00	
6	Suoritusten ajankohta	0	(0=jakson lopussa, 1=jakson alussa)
7	Suoritusten alkuarvo	-10 621,69	=NA(B3;B4;B5;;B6)
8			(engl. PV)

Huomaa, että funktion toiseksi viimeisen argumentin (suoritusten loppuarvo) oletusarvo on 0. Tässä on tyydytty oletusarvoon ja jätetty kyseinen argumentti tyhjäksi (kaksi peräkkäistä puolipistettä). Laskelman mukaan 24 aina jakson lopussa tapahtuvaa 500 euron suoritusta vastaavat 10621,69 euron alkuarvoa.

Jaksollisten suoritusten loppuarvo

Jaksollisten suoritusten loppuarvo voidaan laskea funktiolla **TULEVA.ARVO** (engl. **FV**). Kuvassa näet solussa B7 käytetyn funktion.

	A	B	C
1	Jaksollisten suoritusten loppuarvo		
2			
3	Jakson korkokanta	1,00 %	
4	Jaksoja	24	
5	Suoritus	500,00	
6	Suoritusten ajankohta	0	(0=jakson lopussa, 1=jakson alussa)
7	Suoritusten loppuarvo	-13 486,73	=TULEVA.ARVO(B3;B4;B5;;B6)
8			(engl. FV)

Huomaa, että funktion toiseksi viimeisen argumentin (suoritusten alkuarvo) oletusarvo on 0. Tässä on tyydytty oletusarvoon ja jätetty kyseinen argumentti tyhjäksi (kaksi peräkkäistä puolipistettä). Laskelman mukaan 24 aina jakson lopussa tapahtuvaa 500 euron suoritusta vastaavat 13486,73 euron loppuarvoa.

Jaksollinen suoritus

Jaksollinen suoritus voidaan laskea funktiolla **MAKSU** (engl. **PMT**). Kuvassa näet solussa B8 käytetyn funktion.

	A	B	C
1	Jaksollinen suoritus		
2			
3	Jakson korkokanta	1,00 %	
4	Jaksoja	24	
5	Suoritusten alkuarvo	10 000,00	
6	Suoritusten loppuarvo	0,00	
7	Suoritusten ajankohta	0	(0=jakson lopussa, 1=jakson alussa)
8	Suoritus	-470,73	=MAKSU(B3;B4;B5;B6;B7)
9			(engl. PMT)

Laskelman mukaan 10000 euron alkuarvoa vastaa 24 jakson lopussa tapahtuvaa 470,73 euron suoritusta.

Jos haluat tietää tiettyä loppuarvoa vastaavan suorituksen, niin aseta alkuarvo nolllaksi ja loppuarvo haluamaksesi.

Tarvittaessa sekä alkuarvo että loppuarvo voidaan asettaa nolllasta poikkeaviksi. Jos esimerkiksi haluat tietää, minkälaisilla suorituksilla 10000 euron lainasta saadaan maksettua puolet pois, niin aseta alkuarvoksi 10000 ja loppuarvoksi -5000 (maksamatta jäänyttä alkuarvoa vastaava loppuarvo on erimerkinen kuin alkuarvo).

Jaksolliset suoritukset - korkokanta

Jaksollisten suoritusten korkokanta voidaan laskea funktiolla **KORKO** (engl. **RATE**). Kuvassa näet solussa B8 käytetyn funktion.

	A	B	C
1	Jaksolliset suoritukset - korkokanta		
2			
3	Jaksoja	24	
4	Suoritus	-500,00	
5	Suoritusten alkuarvo	10 000,00	
6	Suoritusten loppuarvo	0,00	
7	Suoritusten ajankohta	0	(0=jakson lopussa, 1=jakson alussa)
8	Jakson korkokanta	1,51 %	=KORKO(B3;B4;B5;B6;B7)
9			(engl. RATE)

Laskelman mukaan 1,51% korkokannalla 500 euron suoritukset vastaavat 10000 euron alkuarvoa.

Jos haluat tietää tiettyä loppuarvoa vastaavan korkokannan, niin aseta alkuarvo nolllaksi ja loppuarvo haluamaksesi.

Tarvittaessa sekä alkuarvo että loppuarvo voidaan asettaa nolllasta poikkeaviksi. Jos esimerkiksi haluat tietää, minkälaisella korkokannalla 10000 euron lainasta saadaan maksettua puolet pois, niin aseta alkuarvoksi 10000 ja loppuarvoksi -5000 (maksamatta jäänyttä alkuarvoa vastaava loppuarvo on erimerkkinen kuin alkuarvo).

Jaksollisten suoritukset - jaksojen lukumäärä

Jaksollisten suoritusten jaksojen lukumäärä voidaan laskea funktiolla **NJAKSO** (engl. **NPER**). Kuvassa näet solussa B8 käytetyn funktion.

	A	B	C
1	Jaksolliset suoritukset - jaksojen lukumäärä		
2			
3	Jakson korkokanta	2,00 %	
4	Suoritus	-500,00	
5	Suoritusten alkuarvo	10 000,00	
6	Suoritusten loppuarvo	0,00	
7	Suoritusten ajankohta	0	(0=jakson lopussa, 1=jakson alussa)
8	Jaksoja	26	=NJAKSO(B3;B4;B5;B6;B7)
9			(engl. NPER)

Laskelman mukaan tarvitaan 26 jaksoa, jotta 500 euron suoritukset vastaavat 10000 euron alkuarvoa.

Jos haluat tietää tiettyä loppuarvoa vastaavan jaksojen lukumäärän, niin aseta alkuarvo nolllaksi ja loppuarvo haluamaksesi.

Tarvittaessa sekä alkuarvo että loppuarvo voidaan asettaa nolllasta poikkeaviksi. Jos esimerkiksi haluat tietää kuinka monella suorituksella 10000 euron lainasta saadaan maksettua puolet pois, niin aseta alkuarvoksi 10000 ja loppuarvoksi -5000 (maksamatta jäänyttä alkuarvoa vastaava loppuarvo on erimerkkinen kuin alkuarvo).

2.3 Luotot

Luoton todellinen vuosikorko

Todellinen vuosikorko on korkokanta, jota käyttäen maksuerien diskontattu arvo on yhtä suuri kuin luoton käteisarvo (luoton määrä vähennettynä nostokuluilla). Jos maksuerien tarkat päivämäärät eivät ole tiedossa, mutta maksuerät maksetaan säännöllisesti, niin todellinen vuosikorko voidaan arvioida funktiolla **SISÄINEN.KORKO** (engl. **IRR**). Huomaa kuitenkin, että maksuerien tarkkoja päivämääriä käyttäen laskettu todellinen vuosikorko saattaa hivenen poiketa saadusta arviosta. Kuvassa näet esimerkin luotosta, joka maksetaan takaisin vuosineljänneksittäin. Kuvaan on merkitty näkyviin solussa B7 käytetty funktio ja solussa B9 käytetty laskukaava. Solussa B9 on muunnettu vuosineljänneksen korko konformiseksi vuosikoroksi. Huomaa, että luoton käteisarvon täytyy olla erimerkkinen kuin maksuerät.

	A	B	C
1	Luoton todellinen vuosikorko - todelliset maksupäivät eivät tiedossa		
2			
3	Seuraavassa esimerkissä oletetaan maksuerien tapahtuvan vuosineljänneksittäin.		
4	Luoton käteisarvo	397 000,00	
5	1. maksuerä	-180 000,00	
6	2. maksuerä	-230 000,00	
7	Vuosineljänneksen korko	2,09 %	=SISÄINEN.KORKO(B4:B6)
8			(engl. IRR)
9	Todellinen vuosikorko	8,62 %	= $(1+B7)^4-1$


Jos maksuerien tarkat päivämäärät ovat tiedossa (maksuerien ei tarvitse olla ajallisesti tasavälisiä), niin todellisen vuosikoron laskemiseksi on kaksi vaihtoehtoista tapaa:

Tapa 1

Kirjoitetaan kaava, jossa maksuerät diskontataan luoton aloitusajankohtaan. Diskonttaukseen käytettävän korkokannan osalta viitataan soluun, johon todellinen vuosikorko halutaan laskea. Korkopäivinä käytetään todellisia päiviä, jotka saadaan päivämäärien vähennyslaskulla. Kuvassa näet soluun B8 kirjoitetun kaavan. Koron osalta viitataan soluun B10.

	A	B	C
1	Luoton todellinen vuosikorko - tarkat maksupäivät tiedossa		
2			
3	TAPA 1 (Tavoitteenhaku / Goal seek)		
4			
5	19.touko	397 000,00	Luoton käteisarvo
6	30.elo	200 000,00	1. maksuerä
7	30.syys	230 000,00	2. maksuerä
8	Maksuerien diskontattu arvo	397000	= $B7/(1+B10)^{(A7-A5)/365}+B6/(1+B10)^{(A6-A5)/365}$
9			
10	Todellinen vuosikorko	27,62 %	

Korko on laskettu soluun B10 **Tiedot - Entä-jos-analyysi - Tavoitteenhaku** -toiminnolla (engl. **Data - What-If Analysis - Goal Seek**). Excel 2003 ja vanhemmissa versioissa **Tavoitteen haku** -toiminto löytyy **Työkalut**-valikosta (engl. **Tools**-valikosta). Tavoitteenhakua on käytetty seuraavin määrittelyin:



Tapa 2

Toinen tapa on käyttää funktiota **SISÄINEN.KORKO.JAKSOTON** (engl. **XIRR**). Jos et löydä funktiota, niin varmista että Excelin analyysityökalut ovat käytössä. Jos käytät analyysityökaluja ensimmäistä kertaa, niin:

1. Napsauta **Office**-painiketta ja valitse alhaalta **Excelin asetukset** (engl. **Excel Options**).
2. Valitse vasemmalta **Apuohjelmat** (engl. **Add Ins**) ja valitse sitten alhaalta **Hallinta**-ruudusta (engl. **Manage**-ruudusta) **Excel-apuohjelmat** (engl. **Excel Add Ins**).
3. Valitse **Siirry** (engl. **Go**).
4. Valitse luettelosta **Analyysityökalut** (engl. **Analysis ToolPak**) ja valitse **OK**. Jos Excel huomauttaa, että analyysityökaluja ei ole asennettu, niin valitse **Kyllä** (engl. **Yes**) asentaaksesi ne.

Excel 2003 ja vanhemmissa versioissa analyysityökalut otetaan käyttöön toiminnolla **Työkalut - Apuohjelmat - Analyysityökalut** (engl. **Tools - Add Ins - Analysis ToolPak**).

Funktiolle annetaan argumentteina maksuerät ja niiden päivämäärät. Edellä tavoitteenhaulla ratkaistu esimerkki lasketaan funktiolla seuraavasti:

	A	B	C
1	Luoton todellinen vuosikorko - tarkat maksupäivät tiedossa		
11			
12	TAPA 2 (Valmista funktiota käyttäen)		
13			
14	19.touko	397 000,00	Luoton käteisarvo
15	30.elo	-200 000,00	1. maksuerä
16	30.syys	-230 000,00	2. maksuerä
17			
18	Todellinen vuosikorko	27,62 %	=SISÄINEN.KORKO.JAKSOTON(B14:B16;A14:A16)
19			<i>(engl. XIRR)</i>

Funktion käytössä on huomioitava, että päivämäärien täytyy olla Excelin tunnistamassa päivämäärämuodossa. Myös vuosiluvun täytyy olla oikein annettu, jotta mahdolliset karkausvuodet tulevat oikein huomioiduiksi (yllä vuosiluku ei ole näkyvillä, koska päivämäärästä on valittu näytettäväksi vain päivä ja kuukausi).

Tasaerä- eli annuiteettiluotto

Tasaeräluoton tasaerä voidaan laskea funktiolla **MAKSU** (engl. **PMT**). Funktion ensimmäisenä argumenttina annettavan koron on oltava maksuväliä vastaava korko. Maksuväliä vastaava korko lasketaan luottojen yhteydessä yleensä relativisesti vastaavasta vuosikorosta. Kuvassa näet soluun B7 kirjoitetun funktion.

	A	B	C
1	Tasaerä- eli annuiteettiluotto		
2			
3	Vuotuinen korko	5,0 %	
4	Maksueriä vuodessa	12	
5	Laina-aika vuosina	15	
6	Lainan määrä	200 000,00 €	
7	Maksuerä	-1 581,59 €	=MAKSU(B3/B4;B4*B5;B6)
8			<i>(engl. PMT)</i>

Maksuerä koostuu korosta ja lyhennyksestä. Laina-ajan alussa koron osuus on suurimmillaan. Laina-ajan edetessä koron osuus pienenee ja lyhennyksen osuus vastaavasti kasvaa. Tiettyyn maksuerään liittyvä korko voidaan laskea **IPMT**-funktiolla ja lyhennys **PPMT**-funktiolla (funktiot samannimiset suomenkielisessä ja englanninkielisessä versiossa). Excel-esimerkeistä löydät esimerkkilainalle kyseisiä funktioita käyttäen laske-
tun lyhennystaulukon.

Leasingvuokra

Leasingvuokra voidaan laskea funktiolla **MAKSU** (engl. **PMT**). Leasingvuokrauksessa on yleistä, että vuokra maksetaan kuukausittain tai neljännesvuosittain etukäteen. Kuvassa näet soluun B7 kirjoitetun funktion. Huomaa, että hankintahinnan ja jäännösarvon täytyy olla funktion argumentteina erimerkkiset. Funktion viimeinen argumentti (1) merkitsee sitä, että maksut suoritetaan etukäteen.

	A	B	C
1	Leasingvuokra		
2			
3	Jakson korkokanta	1,00 %	
4	Leasingkausi jaksoina	36	
5	Hankintahinta	-150 000,00 €	
6	Jäännösarvo	50 000,00 €	
7	Leasingvuokra/jakso	3 783,60 €	=MAKSU(B3;B4;B5;B6;1)
8			<i>(engl. PMT)</i>

Leasingvuokrauksen sisäinen korko

Leasingvuokrauksen sisäinen korko voidaan laskea funktiolla **KORKO** (engl. **RATE**). Funktiota käytettäessä täytyy huomioida, että leasingvuokrauksessa vuokra maksetaan etukäteen (funktion viimeinen argumentti 1). Kuvassa näet soluun B7 kirjoitetun funktion. Huomaa, että hankintahinnan ja jäännösarvon täytyy olla funktion argumentteina erimerkkiset ja leasingvuokran täytyy olla jäännösarvon kanssa samanmerkkinen. Funktio antaa jakson (esim. kuukausi) korkokannan, joka voidaan muuttaa konformiseksi vuosikoroksi.

	A	B	C
1	Leasingin korkokanta		
2			
3	Leasingkausi jaksoina	36	
4	Leasingvuokra/jakso	3 700,00 €	
5	Hankintahinta	-150 000,00 €	
6	Jäännösarvo	50 000,00 €	
7	Jakson korkokanta	0,92 %	=KORKO(B3;B4;B5;B6;1)
8			<i>(engl. RATE)</i>

Jos esimerkissämme jaksona on kuukausi, niin vastaava konforminen vuosikorko on $(1+0,0092)^{12}-1=11,59\%$.

2.4 Investointilaskelmat

Nykyarvomenetelmä

Nykyarvomenetelmää käytettäessä lasketaan investointiin liittyvien tuottojen ja kustannusten nykyarvot. Nykyarvoa laskettaessa pitää valita tilanteeseen sopiva laskentakorkokanta. Investointia voidaan käytettyjen lukujen valossa pitää kannattavana, jos tuottojen nykyarvo on suurempi kuin kustannusten nykyarvo. Investoinnin vuotuiset tuotot ja kustannukset voidaan muuntaa nykyarvoiksi funktiota **NA** (engl. **PV**) käyttäen, jos tuotot ja kustannukset pysyvät joka vuosi samoina. Myös jäännösarvo voidaan muuntaa nykyarvoksi funktiota **NA** (engl. **PV**) käyttäen. Kuvassa näet C-sarakkeeseen kirjoitetut funktiot ja kaavat. Vuotuiset kustannukset, vuotuiset tuotot ja jäännösarvo on lähtötietoina annettu miinus-merkkisinä. Tämän ainoana tarkoituksena on se, että näin kaikki nykyarvot on saatu positiivisina.

	A	B	C	D	E
1	Investointilaskelma - tuottojen ja kustannusten nykyarvojen erotus				
2					
3	Laskentakorkokanta	12 %			
4	Investointiaika vuosina	5			
5					
6			Nykyarvo		
7	Hankintakustannukset	1500000	1500000	=B7	
8	Vuotuiset kustannukset	-120000	432573	=NA(B3;B4;B8)	<i>(engl. FV)</i>
9	Vuotuiset tuotot	-500000	1802388	=NA(B3;B4;B9)	
10	Jäännösarvo	-300000	170228	=NA(B3;B4;B10)	
11					
12	Tuottojen nykyarvo		1972616	=C9+C10	
13	Kustannusten nykyarvo		1932573	=C7+C8	
14	Erotus		40043	=C12-C13	

Jos arvioidut tuotot ja kustannukset vaihtelevat vuodesta toiseen, niin ne voidaan muuntaa yksitellen nykyarvoiksi **NA** (engl. **PV**) käyttäen tai kaikki kerralla funktiolla **NNA** (engl. **NPV**). Kuvassa näet C-sarakkeeseen kirjoitetut kaavat ja funktiot. Soluun C12 on laskettu viiden vuoden nettotuottojen nykyarvo.

	A	B	C	D	E
1	Investointilaskelma - tuottojen ja kustannusten nykyarvojen erotus				
2					
3	Laskentakorkokanta	12 %			
4	Investointiaika (vuotta)	5			
5					
6			Nykyarvo		
7	Hankintakustannukset	1500000	1500000	=B7	
8	1. vuoden nettotuotto	500000			
9	2. vuoden nettotuotto	450000			
10	3. vuoden nettotuotto	400000			
11	4. vuoden nettotuotto	400000			
12	5. vuoden nettotuotto	300000	1514313	=NNA(B3;B8:B12)	(engl. NPV)
13	Jäännösarvo	-300000	170228	=NA(B3;B4;;B13)	(engl. PV)
14					
15	Tuottojen nykyarvo		1684541	=C12+C13	
16	Kustannusten nykyarvo		1500000	=C7	
17	Erotus		184541	=C15-C16	

Annuiteettimenetelmä

Annuiteettimenetelmää käytettäessä lasketaan investointiin liittyviä kustannuksia ja tuottoja (mahdollinen jäännösarvo lasketaan mukaan tuottoihin) vastaavat annuiteetit. Investointia voidaan käytettyjen lukujen valossa pitää kannattavana, jos tuottojen annuiteetti on suurempi kuin kustannusten annuiteetti. Annuiteetteja laskettaessa pitää valita tilanteeseen sopiva laskentakorkokanta. Investoinnin hankintameno ja jäännösarvo muunnetaan annuiteetiksi funktiolla **MAKSU** (engl. **PMT**). Kuvassa näet C-sarakkeeseen kirjoitetut funktiot ja kaavat.

	A	B	C	D	E
1	Investointilaskelma - tulo- ja menoannuiteetti				
2					
3	Laskentakorkokanta	12 %			
4	Investointiaika vuosina	5			
5					
6			Annuiteetti		
7	Hankintakustannukset	1500000	-416 115	=MAKSU(B3;B4;B7)	(engl. PMT)
8	Vuotuiset kustannukset	-120000	-120 000	=B8	
9	Vuotuiset tuotot	500000	500 000	=B9	
10	Jäännösarvo	-300000	47 223	=MAKSU(B3;B4;;B10)	
11					
12	Menoannuiteetti		-536 115	=C7+C8	
13	Tuloannuiteetti		547 223	=C9+C10	

Sisäisen korkokannan menetelmä

Sisäisen korkokannan menetelmässä lasketaan mitä korkokantaa käytettäessä investoinnin tuottojen (mahdollinen jäännösarvo lasketaan mukaan tuottoihin) nykyarvo on yhtä suuri kuin kustannusten nykyarvo. Jos vuotuiset nettotuotot (vuotuisten tuottojen ja kustannusten erotus) pysyvät samoina koko investointiajan, niin sisäinen korkokanta lasketaan funktiolla **KORKO** (engl. **RATE**). Kuvassa näet soluun B7 kirjoitetun funktion.

Huomaa, että funktion argumentteina hankintameno on erimerkkinen kuin vuotuiset tuotot ja jäännösarvo.

	A	B	C
1	Investointilaskelma - sisäinen korkokanta		
2			
3	Investointiaika vuosina	5	
4	Vuotuinen nettotuotto	380000	
5	Investoinnin hankintameno	-1500000	
6	Jäännösarvo	300000	
7	Sisäinen korko	13,0 %	=KORKO(B3;B4;B5;B6)
8			(engl. RATE)

Jos vuotuiset nettotuotot vaihtelevat vuodesta toiseen, niin sisäinen korkokanta lasketaan funktiolla **SISÄINEN.KORKO** (engl. **IRR**). Kuvassa näet soluun B9 kirjoitetun funktion.

	A	B	C
1	Investointilaskelma - sisäinen korkokanta		
2			
3	0. vuosi	-1500000	(hankintameno)
4	1. vuosi	400000	(nettotuotto)
5	2. vuosi	380000	(nettotuotto)
6	3. vuosi	360000	(nettotuotto)
7	4. vuosi	340000	(nettotuotto)
8	5. vuosi	620000	(nettotuotto, joka sisältää jäännösarvon)
9	Sisäinen korko	11,60 %	=SISÄINEN.KORKO(B3:B8)
10			(engl. IRR)